defmodule SportywebWeb.UserRegistrationLive do
  use SportywebWeb, :live_view

  alias Sportyweb.Accounts
  alias Sportyweb.Accounts.User

  def render(assigns) do
    ~H"""
    <div class="mx-auto max-w-md">
      <.header class="text-center">
        Ein Konto erstellen
        <:subtitle>
          Sie haben bereits ein Konto?
          <.link navigate={~p"/users/log_in"} class="font-semibold text-brand hover:underline">
            Jetzt anmelden.
          </.link>
        </:subtitle>
      </.header>

      <.card class="mt-8">
        <.simple_form
          for={@form}
          id="registration_form"
          phx-submit="save"
          phx-change="validate"
          phx-trigger-action={@trigger_submit}
          action={~p"/users/log_in?_action=registered"}
          method="post"
        >
          <.error :if={@check_errors}>
            Bitte überprüfen Sie ihre Eingaben.
          </.error>

          <.input_grids>
            <.input_grid>
              <div class="col-span-12">
                <.input field={@form[:email]} type="email" label="E-Mail-Adresse" required />
              </div>

              <div class="col-span-12">
                <.input field={@form[:password]} type="password" label="Passwort" required />
              </div>
            </.input_grid>
          </.input_grids>

          <:actions>
            <.button phx-disable-with="Konto wird erstellt..." class="w-full">
              Konto erstellen
            </.button>
          </:actions>
        </.simple_form>
      </.card>
    </div>
    """
  end

  def mount(_params, _session, socket) do
    changeset = Accounts.change_user_registration(%User{})

    socket =
      socket
      |> assign(trigger_submit: false, check_errors: false)
      |> assign_form(changeset)

    {:ok, socket, temporary_assigns: [form: nil]}
  end

  def handle_event("save", %{"user" => user_params}, socket) do
    case Accounts.register_user(user_params) do
      {:ok, user} ->
        {:ok, _} =
          Accounts.deliver_user_confirmation_instructions(
            user,
            &url(~p"/users/confirm/#{&1}")
          )

        changeset = Accounts.change_user_registration(user)
        {:noreply, socket |> assign(trigger_submit: true) |> assign_form(changeset)}

      {:error, %Ecto.Changeset{} = changeset} ->
        {:noreply, socket |> assign(check_errors: true) |> assign_form(changeset)}
    end
  end

  def handle_event("validate", %{"user" => user_params}, socket) do
    changeset = Accounts.change_user_registration(%User{}, user_params)
    {:noreply, assign_form(socket, Map.put(changeset, :action, :validate))}
  end

  defp assign_form(socket, %Ecto.Changeset{} = changeset) do
    form = to_form(changeset, as: "user")

    if changeset.valid? do
      assign(socket, form: form, check_errors: false)
    else
      assign(socket, form: form)
    end
  end
end
